# $Id: PKGBUILD 183374 2013-04-21 22:09:43Z heftig $
# Maintainer: Ionut Biru <ibiru@archlinux.org>

pkgname=gnome-contacts
pkgver=3.8.1
pkgrel=1
pkgdesc="Contacts Manager for GNOME"
arch=(i686 x86_64)
url="http://www.gnome.org"
license=('GPL2')
groups=(gnome)
depends=('gtk3' 'folks' 'libnotify' 'gnome-desktop' 'dconf' 'gnome-online-accounts' 'libgee' 'telepathy-glib' 'cheese')
makedepends=('intltool' 'vala' 'gobject-introspection')
install=$pkgname.install
source=(http://download.gnome.org/sources/$pkgname/${pkgver%.*}/$pkgname-$pkgver.tar.xz)
sha256sums=('c9b2ffba754f4e1e37788195c814b12dfbf79b35042cd811b8690f890572edcb')

build() {
  cd $pkgname-$pkgver
  ./configure --prefix=/usr --libexecdir=/usr/lib/gnome-contacts
  make
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
}

# vim:set ts=2 sw=2 et:
